

NAG Library Routine Document

G13ASF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G13ASF is a diagnostic checking routine suitable for use after fitting a Box–Jenkins ARMA model to a univariate time series using G13AEF or G13AFF. The residual autocorrelation function is returned along with an estimate of its asymptotic standard errors and correlations. Also, G13ASF calculates the Box–Ljung portmanteau statistic and its significance level for testing model adequacy.

2 Specification

```

SUBROUTINE G13ASF (N, V, MR, M, PAR, NPAR, ISHOW, R, RCM, LDRCM, CHI,      &
                  IDF, SIGLEV, IW, LIW, WORK, LWORK, IFAIL)
INTEGER           N, MR(7), M, NPAR, ISHOW, LDRCM, IDF, IW(LIW), LIW,    &
                  LWORK, IFAIL
REAL (KIND=nag_wp) V(N), PAR(NPAR), R(M), RCM(LDRCM,M), CHI, SIGLEV,   &
                  WORK(LWORK)

```

3 Description

Consider the univariate multiplicative autoregressive-moving average model

$$\phi(B)\Phi(B^s)(W_t - \mu) = \theta(B)\Theta(B^s)\epsilon_t \quad (1)$$

where W_t , for $t = 1, 2, \dots, n$, denotes a time series and ϵ_t , for $t = 1, 2, \dots, n$, is a residual series assumed to be normally distributed with zero mean and variance σ^2 (> 0). The ϵ_t 's are also assumed to be uncorrelated. Here μ is the overall mean term, s is the seasonal period and B is the backward shift operator such that $B^r W_t = W_{t-r}$. The polynomials in (1) are defined as follows:

$$\phi(B) = 1 - \phi_1 B - \phi_2 B^2 - \dots - \phi_p B^p$$

is the non-seasonal autoregressive (AR) operator;

$$\theta(B) = 1 - \theta_1 B - \theta_2 B^2 - \dots - \theta_q B^q$$

is the non-seasonal moving average (MA) operator;

$$\Phi(B^s) = 1 - \Phi_1 B^s - \Phi_2 B^{2s} - \dots - \Phi_P B^{Ps}$$

is the seasonal AR operator; and

$$\Theta(B^s) = 1 - \Theta_1 B^s - \Theta_2 B^{2s} - \dots - \Theta_Q B^{Qs}$$

is the seasonal MA operator. The model (1) is assumed to be stationary, that is the zeros of $\phi(B)$ and $\Phi(B^s)$ are assumed to lie outside the unit circle. The model (1) is also assumed to be invertible, that is the zeros of $\theta(B)$ and $\Theta(B^s)$ are assumed to lie outside the unit circle. When both $\Phi(B^s)$ and $\Theta(B^s)$ are absent from the model, that is when $P = Q = 0$, then the model is said to be non-seasonal.

The estimated residual autocorrelation coefficient at lag l , \hat{r}_l , is computed as:

$$\hat{r}_l = \frac{\sum_{t=l+1}^n (\hat{\epsilon}_{t-l} - \bar{\epsilon})(\hat{\epsilon}_t - \bar{\epsilon})}{\sum_{t=1}^n (\hat{\epsilon}_t - \bar{\epsilon})^2}, \quad l = 1, 2, \dots$$

where $\hat{\epsilon}_t$ denotes an estimate of the t th residual, ϵ_t , and $\bar{\epsilon} = \sum_{t=1}^n \hat{\epsilon}_t/n$. A portmanteau statistic, $Q_{(m)}$, is calculated from the formula (see Box and Ljung (1978)):

$$Q_{(m)} = n(n+2) \sum_{l=1}^m \hat{r}_l^2 / (n-l)$$

where m denotes the number of residual autocorrelations computed. (Advice on the choice of m is given in Section 9.2.) Under the hypothesis of model adequacy, $Q_{(m)}$ has an asymptotic χ^2 -distribution on $m - p - q - P - Q$ degrees of freedom. Let $\hat{r}^T = (\hat{r}_1, \hat{r}_2, \dots, \hat{r}_m)$ then the variance-covariance matrix of \hat{r} is given by:

$$\text{Var}(\hat{r}) = [I_m - X(X^T X)^{-1} X^T] / n.$$

The construction of the matrix X is discussed in McLeod (1978). (Note that the mean, μ , and the residual variance, σ^2 , play no part in calculating $\text{Var}(\hat{r})$ and therefore are not required as input to G13ASF.)

Note: for additive models with fixed parameter values (i.e., fitted by G13DDF) G13DSF should be used instead of G13ASF.

4 References

Box G E P and Ljung G M (1978) On a measure of lack of fit in time series models *Biometrika* **65** 297–303

McLeod A I (1978) On the distribution of the residual autocorrelations in Box–Jenkins models *J. Roy. Statist. Soc. Ser. B* **40** 296–302

5 Arguments

1: N – INTEGER *Input*

On entry: n , the number of observations in the residual series.

If G13ASF is used following a call to G13AEF, then N must be the value ICOUNT(2) returned by G13AEF.

If G13ASF is used following a call to G13AFF, then N must be the value NRES returned by G13AFF.

Constraint: $N \geq 3$.

2: V(N) – REAL (KIND=nag_wp) array *Input*

On entry: $V(t)$ must contain an estimate of ϵ_t , for $t = 1, 2, \dots, n$.

If G13ASF is used following a call to G13AEF then the actual argument V must be EXR(ICOUNT(1) + 1) as returned by G13AEF.

If G13ASF is used following a call to G13AFF then the actual argument V must be RES as returned by G13AFF.

Constraint: V must contain at least two distinct elements.

3: MR(7) – INTEGER array *Input*

On entry: the orders vector (p, d, q, P, D, Q, s) as supplied to G13AEF or G13AFF.

Constraints:

$$\begin{aligned} p, q, P, Q, s &\geq 0; \\ p + q + P + Q &> 0; \end{aligned}$$

if $s = 0$, then $P = 0$ and $Q = 0$.

- 4: M – INTEGER *Input*
On entry: the value of m , the number of residual autocorrelations to be computed. See Section 9.2 for advice on the value of M.
Constraint: $\text{NPAR} < M < N$.
- 5: PAR(NPAR) – REAL (KIND=nag_wp) array *Input*
On entry: the parameter estimates in the order $\phi_1, \phi_2, \dots, \phi_p, \theta_1, \theta_2, \dots, \theta_q, \Phi_1, \Phi_2, \dots, \Phi_P, \Theta_1, \Theta_2, \dots, \Theta_Q$ only.
Constraint: the elements in PAR must satisfy the stationarity and invertibility conditions.
- 6: NPAR – INTEGER *Input*
On entry: the total number of ϕ, θ, Φ and Θ parameters, i.e., $\text{NPAR} = p + q + P + Q$.
Constraint: $\text{NPAR} = \text{MR}(1) + \text{MR}(3) + \text{MR}(4) + \text{MR}(6)$.
- 7: ISHOW – INTEGER *Input*
On entry: must be nonzero if the residual autocorrelations, their standard errors and the portmanteau statistics are to be printed and zero otherwise.
 These quantities are available also as output variables in R, RCM, CHI, IDF and SIGLEV.
- 8: R(M) – REAL (KIND=nag_wp) array *Output*
On exit: an estimate of the residual autocorrelation coefficient at lag l , for $l = 1, 2, \dots, m$. If $\text{IFAIL} = 3$ on exit then all elements of R are set to zero.
- 9: RCM(LDRCM, M) – REAL (KIND=nag_wp) array *Output*
On exit: the estimated standard errors and correlations of the elements in the array R. The correlation between $R(i)$ and $R(j)$ is returned as $\text{RCM}(i, j)$ except that if $i = j$ then $\text{RCM}(i, j)$ contains the standard error of $R(i)$. If on exit, $\text{IFAIL} \geq 5$, then all off-diagonal elements of RCM are set to zero and all diagonal elements are set to $1/\sqrt{n}$.
- 10: LDRCM – INTEGER *Input*
On entry: the first dimension of the array RCM as declared in the (sub)program from which G13ASF is called.
Constraint: $\text{LDRCM} \geq M$.
- 11: CHI – REAL (KIND=nag_wp) *Output*
On exit: the value of the portmanteau statistic, $Q_{(m)}$. If $\text{IFAIL} = 3$ on exit then CHI is returned as zero.
- 12: IDF – INTEGER *Output*
On exit: the number of degrees of freedom of CHI.
- 13: SIGLEV – REAL (KIND=nag_wp) *Output*
On exit: the significance level of CHI based on IDF degrees of freedom. If $\text{IFAIL} = 3$ on exit, SIGLEV is returned as one.

- 14: IW(LIW) – INTEGER array *Workspace*
 15: LIW – INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which G13ASF is called.

Constraint: $LIW \geq \max(MR(1), MR(3), MR(4), MR(6))$.

- 16: WORK(LWORK) – REAL (KIND=nag_wp) array *Workspace*
 17: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13ASF is called.

Constraint: $LWORK \geq NPAR \times (M + NPAR + 1) + \max(MR(1), MR(3), MR(4), MR(6)) \times \max(MR(7), 1) + M$.

- 18: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $IFAIL \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G13ASF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, MR(1) < 0,
 or MR(3) < 0,
 or MR(4) < 0,
 or MR(6) < 0,
 or MR(7) < 0,
 or MR(7) = 0 and either MR(4) > 0 or MR(6) > 0,
 or MR(1) = MR(3) = MR(4) = MR(6) = 0,
 or $M \leq NPAR$,
 or $M \geq N$,
 or $N < 3$,
 or $NPAR \neq MR(1) + MR(3) + MR(4) + MR(6)$,
 or $LDRCM < M$,
 or LIW is too small,
 or LWORK is too small.

IFAIL = 2

On entry, the autoregressive (or moving average) parameters are extremely close to or outside the stationarity (or invertibility) region. To proceed, you must supply different parameter estimates in the array PAR.

IFAIL = 3

On entry, the residuals are practically identical giving zero (or near zero) variance. In this case CHI is set to zero and SIGLEV to one and all the elements of R are set to zero.

IFAIL = 4

This is an unlikely exit brought about by an excessive number of iterations being needed to evaluate the zeros of the AR or MA polynomials. All output arguments are undefined.

IFAIL = 5

On entry, one or more of the AR operators has a factor in common with one or more of the MA operators. To proceed, this common factor must be deleted from the model. In this case, the off-diagonal elements of RCM are returned as zero and the diagonal elements set to $1/\sqrt{n}$. All other output quantities will be correct.

IFAIL = 6

This is an unlikely exit. At least one of the diagonal elements of RCM was found to be either negative or zero. In this case all off-diagonal elements of RCM are returned as zero and all diagonal elements of RCM set to $1/\sqrt{n}$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computations are believed to be stable.

8 Parallelism and Performance

G13ASF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G13ASF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

9.1 Timing

The time taken by G13ASF depends upon the number of residual autocorrelations to be computed, m .

9.2 Choice of m

The number of residual autocorrelations to be computed, m should be chosen to ensure that when the ARMA model (1) is written as either an infinite order autoregressive process:

$$W_t - \mu = \sum_{j=1}^{\infty} \pi_j (W_{t-j} - \mu) + \epsilon_t$$

or as an infinite order moving average process:

$$W_t - \mu = \sum_{j=1}^{\infty} \psi_j \epsilon_{t-j} + \epsilon_t$$

then the two sequences $\{\pi_1, \pi_2, \dots\}$ and $\{\psi_1, \psi_2, \dots\}$ are such that π_j and ψ_j are approximately zero for $j > m$. An overestimate of m is therefore preferable to an under-estimate of m . In many instances the choice $m = 10$ will suffice. In practice, to be on the safe side, you should try setting $m = 20$.

9.3 Approximate Standard Errors

When IFAIL = 5 or 6 all the standard errors in RCM are set to $1/\sqrt{n}$. This is the asymptotic standard error of \hat{r}_l when all the autoregressive and moving average parameters are assumed to be known rather than estimated.

9.4 Alternative Applications

G13ASF may be used for diagnostic checking of suitable univariate ARMA models, as described in Section 3, fitted by G13BEF or G13DDF. Great care must be taken in obtaining the input values for G13ASF from the output values from G13BEF or G13DDF.

10 Example

This example fits an ARIMA(1, 1, 2) model to a series of 30 observations. 10 residual autocorrelations are computed.

10.1 Program Text

```

Program g13asfe

!      G13ASF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g13aff, g13asf, nag_wp, x04abf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: iset = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: c, chi, s, siglev
Integer                    :: idf, ifail, ires, ishow, itc, kfc, &
                             kpiv, ldcm, ldrcm, liw, lwork, m, n, &
                             nadv, ndf, nit, npar, nppc, nst, nx, &
                             pp, qp
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: cm(:,,:), par(:), r(:), rcm(:,,:), &
                             sd(:), st(:), v(:), work(:), x(:)
Integer                        :: isf(4), mr(7)
Integer, Allocatable          :: iw(:)
!      .. Intrinsic Procedures ..
Intrinsic                    :: max
!      .. Executable Statements ..
Write (nout,*) 'G13ASF Example Program Results'
Write (nout,*)

```

```

      Flush (nout)

!      Skip heading in data file
      Read (nin,*)

!      Read in problem size
      Read (nin,*) nx, kfc, c

!      Read in the orders
      Read (nin,*) mr(1:7)

!      Calculate NPAR and various array sizes
      npar = mr(1) + mr(3) + mr(4) + mr(6)
      nppc = npar + kfc
      ldcm = nppc
      pp = mr(1) + mr(4)*mr(7)
      qp = mr(3) + mr(6)*mr(7)
      ires = 15*qp + 11*nx + 13*nppc + 8*pp + 12 + 2*(qp+nppc)**2

      Allocate (par(npar),x(nx),sd(nppc),cm(ldcm,nppc),st(nx),v(ires))

!      Read in data
      Read (nin,*) x(1:nx)

!      Read in initial values
      Read (nin,*) par(1:npar)

!      Read in control parameters
      Read (nin,*) kpiv, nit

!      Set the advisory channel to NOUT for monitoring information
      If (kpiv/=0) Then
        nadv = nout
        Call x04abf(iset,nadv)
      End If

!      Fit ARIMA model
      ifail = -1
      Call g13aff(mr,par,npar,c,kfc,x,nx,s,ndf,sd,nppc,cm,ldcm,st,nst,kpiv,      &
        nit,itc,isf,v,ires,n,ifail)
      If (ifail/=0) Then
        If (ifail/=9) Then
          Go To 100
        End If
      End If

!      Read in parameters for G13ASF
      Read (nin,*) m, ishow

      ldrcm = m
      liw = max(mr(1),mr(3),mr(4),mr(6))
      lwork = npar*(n+npar+1) + liw*max(mr(7),1) + m
      Allocate (r(m),rcm(ldrcm,m),iw(liw),work(lwork))

!      Set the advisory channel to NOUT for monitoring information
!      if it has not been set previously
      If (ishow/=0 .And. kpiv==0) Then
        nadv = nout
        Call x04abf(iset,nadv)
      End If

!      Perform diagnostic checks
      ifail = 0
      Call g13asf(n,v,mr,m,par,npar,ishow,r,rcm,ldrcm,chi,idf,siglev,iw,liw,      &
        work,lwork,ifail)

100  Continue

      End Program g13asfe

```

10.2 Program Data

G13ASF Example Program Data

```

30 1 0.0                               : NX,KFC,C
1 1 2 0 0 0 0                          : MR
-217.0 -177.0 -166.0 -136.0 -110.0 -95.0 -64.0 -37.0
-14.0 -25.0 -51.0 -62.0 -73.0 -88.0 -113.0 -120.0
-83.0 -33.0 -19.0 21.0 17.0 44.0 44.0 78.0
 88.0 122.0 126.0 114.0 85.0 64.0      : End of X
0.0 0.0 0.0                             : PAR (initial values)
0 50                                      : KPIV,NIT
10 1                                      : M,ISHOW

```

10.3 Program Results

G13ASF Example Program Results

RESIDUAL AUTOCORRELATION FUNCTION

```

-----
LAG K      1      2      3      4      5      6      7
R(K)      0.020 -0.040 -0.019  0.068 -0.143 -0.046 -0.205
ST.ERROR  0.007  0.125  0.128  0.150  0.168  0.168  0.178
-----
LAG K      8      9     10
R(K)     -0.108 -0.001 -0.058
ST.ERROR  0.179  0.181  0.183
-----

```

```

BOX - LJUNG PORTMANTEAU STATISTIC =      3.465
      SIGNIFICANCE LEVEL =      0.839
(BASED ON 7 DEGREES OF FREEDOM)

```

```

VALUE OF IFAIL PARAMETER ON EXIT FROM G13ASF = 0

```
